

Address	Code	Label	Mnemonic	Comment
---------	------	-------	----------	---------

Anschlüsse an Leiterplatte für Tasten und Anzeige

Adresse	Signal	Anschluss	Funktion
Status, Bit 0	F0	Ein-Aus 15	H= Addresszähler Reset auf 00
Status, Bit 1	F1	Ein-Aus 2	Strobe für Daten lesen, Bit-Selektion an P806-0 bis 2
Status, Bit 2	F2	Ein-Aus 1	Hi = Programmierspannung Ein, /G2 = VPP_SW, Verzögerung ~130uS
Status, Bit 5	SB	Ein-Aus 10	Daten-Eingang nach Strobe an F1
0803	0803-Bit 7	Ein-Aus 3	Addresszähler Takt, 00-FF
0807	DB2	Ein-Aus 12	DB4=H, DB2=H: (14) VPP=50V-ZD2 (30V)
0807	DB4	Ein-Aus 13	DB4=L, DB2=H: (04) VPP=50V-ZD2 (30V) + 4*0.7V DB4=X, DB2=L: (X0) VPP=50V-ZD2 (30V) + 8*0.7V
0806	P806-0	Drucker 16	Datenbit 0, für Programmieren und Lese-Bit Auswahl
	P806-1	Drucker 15	Datenbit 1, für Programmieren und Lese-Bit Auswahl
	P806-2	Drucker 14	Datenbit 2, für Programmieren und Lese-Bit Auswahl
	P806-3	Drucker 13	Datenbit 3, für Programmieren
	P806-4	Drucker 12	Datenbit 4, für Programmieren
	P806-5	Drucker 11	Datenbit 5, für Programmieren
	P806-6	Drucker 10	Datenbit 6, für Programmieren
	P806-7	Drucker 9	Datenbit 7, für Programmieren

Bedienung

Programm starten (Sprung nach 0600, ROM C)				
0420	C4 05		LDI #05	P1=05FF
0422	35		XPAH1	
0423	C4 FF		LDI #FF	
0425	31		XPAL1	
0426	3D		XPPC1	
[GO] (Anzeige bleibt unverändert)				
Alternativ Sprung nach 0600 nur bei aktiver SB Taste, als Anfang des Anwenderprogramms				
0420	06		CSA	Sprung zum Anwenderprogramm bei 042C falls SENSE B bei GO nicht gedrückt
0421	D4 20		ANI #20	
0423	98 07		JZ ANWPRG	
0425	C4 05		LDI #05	Sprung ins Programmier-Programm bei aktiver SENSE B Taste
0427	35		XPAH1	
0428	C4 FF		LDI #FF	
042A	31		XPAL1	
042B	3D		XPPC1	
[GO] mit SENSE B (Anzeige bleibt unverändert)				
042C	xx xx	ANWPRG:	xxxx	Start Anwenderprogramm
Eingabe der Start- und Endadressen				
041C	xx			Startadresse HOB
041D	xx			Startadresse LOB
041E	xx			Endadresse HOB
041F	xx			Endadresse LOB
Tastenfunktionen im PROM Programmer				
O..F, AD, ME, IN	Eingabe von RAM Start- und Endadressen. Funktion analog Betriebsprogramm.			
PT	Start Programmiervorgang. Display wird dunkel, starten mit SENSE A. Programmierung startet immer bei PROM-Adresse 000 oder 100, abhängig von der Schalterstellung auf dem PROM Programmer.			

Address	Code	Label	Mnemonic	Comment
PC				Displacement-Berechnung - Display zeigt 0000 00 - Startadresse eingeben, für Sprünge Adresse des Sprungbefehls + 2 - Mit [IN] übernehmen, Display zeigt wieder 0000 00 - Zieladresse eingeben - Beliebige Funktionstaste, ausser IN - Display zeigt erforderliches Displacement - Wenn Drucker angeschlossen, Ausdruck [Start] -> [Ziel] : V R [Displacement] wobei V=Vorwärts, R=Rückwärts. Erscheint ein F, ist die Eingabe ungültig, Differenz zwischen Start und Ziel > 127 oder kleiner -128 - Erneute Berechnung oder Abbruch mit [RESET]

Konstanten

	0E	KEYIN		Code Taste IN
--	----	-------	--	---------------

Scratchpad

0400	00	BFPCH		Buffer PC HOB
0401	01	BFPCL		Buffer PC LOB
0402	02	BFP1H		Buffer P1 HOB
0403	03	BFP1L		Buffer P1 LOB
0404	04	BFP2H		Buffer P2 HOB
0403	05	BFP2L		Buffer P2 LOB
0410	10	ISEDH		Eingabe SED HOB
0411	11	ISEDL		Eingabe SED LOB
0412	12	DSADH		Anzeige Adresse HOB
0413	13	DSADL		Anzeige Adresse LOB
0414	14	DSCOD		Anzeige Code
0415	15	DEB01		Entprellen1
0416	16	SPKEY		Funktionstaste
0417	17	DEB02		Entprellen2
0419	19	KYLI 2		Tasten-Zeilen2
041A	1A	KYLI 3		Tasten-Zeilen3
041B	1B	KYLI 4		Tasten-Zeilen4
041C	1C	FRADH		Start Adresse HOB
041D	1D	FRADL		Start Adresse LOB
041E	1E	TOADH		End Adresse HOB
041F	1F	TOADL		End Adresse LOB

Einsprung Programmer ROM

0600	C4 01	START:	LDI #01	Sprung nach 0110 (ROM A, EINAU)
0602	37		XPAH3	
0603	C4 0F		LDI #0F	
0605	33		XPAL3	
0606	3F		XPPC3	

Auf Funktionstaste warten

0607	C4 04		LDI #04	P1=0400 (Pointer Scratchpad)
0609	35		XPAH1	
060A	C4 00		LDI #00	
060C	31		XPAL1	
060D	C1 16		LD SPKEY(1)	A=Code Funktionstaste
060F	98 EF		JZ START	Keine, erneut lesen

Funktionstaste auswerten

0611	01		XAE	EX=Tastencode, 01-0E
------	----	--	-----	----------------------

Address	Code	Label	Mnemonic	Comment
0612	C4 06		LDI #06	P2=0622, JMPTAB-1
0614	36		XPAH2	
0615	C4 22		LDI #22	
0617	32		XPAL2	
0618	C2 80		LD 80(2)	A=P2(EX), HOB Sprungziel
061A	37		XPAH3	
061B	C6 0E		LD @0E(2)	P2 + 0E (zeigt auf LOB des Sprungziels)
061D	C2 80		LD 80(2)	A=P2(EX), LOB Sprungziel
061F	33		XPAL3	
0620	3F		XPPC3	Sprung je nach Taste
0621	90 DD		JMP START	Endlosschleife

Tasten-Sprungtabelle

0623	07	JMPTAB:		HOB Taste 01: Adresse = 0743 (PC)
0624	00			HOB Taste 02: Adresse = 0000 (P1)
0625	00			HOB Taste 03: Adresse = 0000 (P2)
0626	00			HOB Taste 04: Adresse = 0000 (P3)
0627	00			HOB Taste 05: Adresse = 0000 (AC)
0628	00			HOB Taste 06: Adresse = 0000 (EX)
0629	00			HOB Taste 07: Adresse = 0000 (BP)
062A	00			HOB Taste 08: Adresse = 0000 (0801)
062B	06			HOB Taste 09: Adresse = 068F (PT)
062C	00			HOB Taste 0A: Adresse = 0000 (SI)
062D	00			HOB Taste 0B: Adresse = 0000 (GO)
062E	06			HOB Taste 0C: Adresse = 0656 (AD)
062F	06			HOB Taste 0D: Adresse = 063E (ME)
0630	06			HOB Taste 0E: Adresse = 067D (IN)
0631	43			LOB Taste 01: Adresse = 0743 (PC)
0632	00			LOB Taste 02: Adresse = 0000 (P1)
0633	00			LOB Taste 03: Adresse = 0000 (P2)
0634	00			LOB Taste 04: Adresse = 0000 (P3)
0635	00			LOB Taste 05: Adresse = 0000 (AC)
0636	00			LOB Taste 06: Adresse = 0000 (EX)
0637	00			LOB Taste 07: Adresse = 0000 (BP)
0638	00			LOB Taste 08: Adresse = 0000 (0801)
0639	8F			LOB Taste 09: Adresse = 068F (PT)
063A	00			LOB Taste 0A: Adresse = 0000 (SI)
063B	00			LOB Taste 0B: Adresse = 0000 (GO)
063C	56			LOB Taste 0C: Adresse = 0656 (AD)
063D	3E			LOB Taste 0D: Adresse = 063E (ME)
063E	7D			LOB Taste 0E: Adresse = 067D (IN)

Taste ME: Nächste Adresse anzeigen

063F	02	KEYME:	CCL	CY=0
0640	C1 13		LD DSADL(1)	A=(Anzeige Adresse LOB)
0642	F4 01		ADI #01	A=A+1+CY
0644	C9 13		ST DSADL(1)	A->(Anzeige Adresse LOB)
0646	C1 12		LD DSADH(1)	A=(Anzeige Adresse HOB)
0648	F4 00		ADI #00	A=A+CY
064A	C9 12		ST DSADH(1)	A->(Anzeige Adresse HOB)
064C	C1 13	LO64C:	LD DSADL(1)	A=(Anzeige Adresse LOB)
044E	32		XPAL2	
044F	C1 12		LD DSADH(1)	A=(Anzeige Adresse HOB)
0651	36		XPAH2	
0652	C2 00		LD 00(2)	A=(Anzeige Adresse)

Address	Code	Label	Mnemonic	Comment
0654	C9 14		ST DSCOD(1)	A->(Anzeige Code)
0656	3F		XPPC3	Return

Taste AD: Vorherige oder eingegebene Adresse anzeigen

0657	C1 10	KEYAD:	LD ISEDH(1)	A=(Eingabe SED HOB)
0659	9C 04		JZ L065F	
065B	C1 11		LD ISEDL(1)	A=(Eingabe SED LOB)
065D	9A 10		JZ ISEDH(2)	
065F	C1 11	L065F:	LD ISEDL(1)	A=(Eingabe SED LOB)
0661	C9 13		ST DSADL(1)	A->(Anzeige Adresse LOB)
0663	C1 10		LD ISEDH(1)	A=(Eingabe SED HOB)
0665	C9 12		ST DSADH(1)	A->(Anzeige Adresse HOB)
0667	C4 00	L0667:	LDI #00	
0669	C9 11		ST ISEDL(1)	0->(Eingabe SED LOB)
066B	C9 10		ST ISEDH(1)	0->(Eingabe SED HOB)
066D	90 DD		JMP L064C	P2=(Anzeige Adresse), Code anzeigen
066F	03		SCL	CY=1
0670	C1 13		LD DSADL(1)	A=(Anzeige Adresse LOB)
0672	FC 01		CAI 01	A+= /01 + CY
0674	C9 13		ST DSADL(1)	A->(Anzeige Adresse LOB)
0676	C1 12		LD DSADH(1)	A=(Anzeige Adresse HOB)
0678	FC 00		CAI 00	A+= /00 +CY
067A	C9 12		ST DSADH(1)	A->(Anzeige Adresse HOB)
067C	90 CE		JMP L064C	P2=(Anzeige Adresse), Code anzeigen

Taste IN: Eingegebenen Code speichern

067E	C1 12	KEYIN:	LD DSADH(1)	P2=(Anzeige Adresse)
0680	36		XPAH2	
0681	C1 13		LD DSADL(1)	
0683	32		XPAL2	
0684	C1 11		LD ISEDL(1)	A=(Eingabe SED LOB)
0686	CE 01		ST @01(2)	A->(P2), P2++
0688	36		XPAH2	
0689	C9 12		ST DSADH(1)	A->(Anzeige Adresse HOB)
068B	32		XPAL2	
068C	C9 13		ST DSADL(1)	A->(Anzeige Adresse LOB)
068E	90 D7		JMP L0667	

Taste PT: PROM programmieren

0690	C4 01	KEYPT:	LDI #01	
0692	07		CAS	F0 setzen, Adresszähler=0
0693	36	SAWAIT:	CSA	
0694	D4 10		ANI #10	SENSE A maskieren
0696	98 FB		JZ SAWAIT	Warten bis SENSE A gedrückt
0698	C4 04		LDI #04	P1=Pointer auf 0400, Scratchpad
069A	35		XPAH1	
069B	C4 00		LDI #00	
069D	31		XPAL1	
069E	C4 08		LDI #08	P2=0800
06A0	36		XPAH2	
06A1	C4 00		LDI #00	
06A3	32		XPAL2	

Address	Code	Label	Mnemonic	Comment
06A4	C1 1C		LD FRADH(1)	P3=(Startadresse)
06A6	37		XPAH3	
06A7	C1 1D		LD FRADL(1)	
06A9	33		XPAL3	
Test ob Startadresse = Endadresse, wenn ja, Rücksprung ins Betriebsprogramm				
06AA	C2 03	L06AA:	LD 03(2)	Adresse 0803 = Adressen-Clock
06AC	C4 00		LDI #00	
06AE	07		CAS	Alle Flags=0
06AF	37		XPAH3	A=(Startadresse HOB)
06B0	01		XAE	EX=(Startadresse HOB)
06B1	40		LDE	A=(Startadresse HOB)
06B2	37		XPAH3	P3H=(Startadresse HOB)
06B3	40		LDE	A=(Startadresse HOB)
06B4	E1 1E		XOR TOADH(1)	
06B6	9C 13		JNZ PROG	Sprung wenn Start HOB <> Ende HOB
06B8	33		XPAL3	A=(Startadresse LOB)
06B9	01		XAE	EX=(Startadresse LOB)
06BA	40		LDE	A=(Startadresse LOB)
06BB	33		XPAL3	P3L=(Startadresse LOB)
06BC	40		LDE	A=(Startadresse LOB)
06BD	E1 1F		XOR TOADL(1)	
06BF	9C 0A		JNZ PROG	Sprung wenn Start LOB <> Ende LOB
Startadresse = Endadresse, Rücksprung ins Betriebsprogramm				
06C1	C4 01		LDI #01	Adresszähler = 0
06C3	07		CAS	
06C4	C4 00	RESET:	LDI #00	Rücksprung nach 0000, Reset
06C6	37		XPAH3	
06C7	C4 00		LDI #00	
06C9	33		XPAL3	
06CA	3F		XPPC3	
Nächstes Byte programmieren: P1=Scratchpad, P2=0800, P3=aktuelle Adresse				
06CB	C7 01	PROG:	LD @01(3)	(Startadresse ++), A irrelevant
06CD	C4 FF		LDI #FF	
06CF	C9 1B		ST KYLI4(1)	Bitzähler = FF
06D1	C4 80		LDI #80	
06D3	C9 1A		ST KYLI3(1)	Brennbit = 80
06D5	90 05		JMP L06DC	
06D7	C1 1A	L06D7:	LD KYLI3(1)	Brennbit >>= 1
06D9	1C		SR	
06DA	C9 1A		ST KYLI3(1)	
06DC	A9 1B	L06DC:	ILD KYLI4(1)	A= (++ Bitzähler)
06DE	D4 08		ANI #08	
06E0	9C C8		JNZ L06AA	Byte fertig, testen ob noch mehr
06E2	C1 1A		LD KYLI3(1)	A=Brennbit
06E4	D3 FF		AND FF(3)	Vergleich mit Bit des Datenbyte
06E6	9C EF		JNZ L06D7	Bit ist gleich, nicht brennen
Nächstes Bit programmieren, in KYLI3 steht zu brennendes Bit				
06E8	C4 00		LDI #00	
06EA	C9 19		ST KYLI2(1)	Brennzähler = 00

Address	Code	Label	Mnemonic	Comment
06EC	07		CAS	Alle Flags löschen
06ED	A9 19	L06ED:	ILD KYLI 2(1)	A=(++ Brennzähler), auch Neuversuch
06EF	D4 0C		ANI #0C	VPP Selektierung maskieren
06F1	1C		SR	Nach Bit 0,1 schieben
06F2	1C		SR	
06F3	F4 03		ADI #03	Offset zu 06FA, Programmierspannungen
06F5	01		XAE	EX=Offset für PC-relatives Laden
06F6	C0 80		LD 80	A=Programmierspannung (EX)
06F8	90 04		JMP L06FE	Bit programmieren
06FA	10		DB 10	EX=3: VPP=35.6V
06FB	04		DB 04	EX=4: VPP=32.8V
06FC	14		DB 14	EX=5: VPP=30V
06FD	14		DB 14	EX=6: VPP=30V
06FE	CA 07	L06FE:	ST 07(2)	(807)=A, Selektieren der VPP
0700	C4 04		LDI #04	F2=1: Programmierspannung Ein
0702	07		CAS	
0703	C1 1A		LD KYLI 3(1)	A=Brennbit
0705	08		NOP	
0706	CA 06		ST 06(2)	Bit an Datenport ausgeben
0708	01		XAE	
0709	C4 00		LDI #00	Datenport = 0
070B	CA 06		ST 06(2)	
070D	07		CAS	F2=0: Programmierspannung Aus
070E	8F 01		DLY 01	
0710	C1 1B		LD KYLI 4(1)	A=Bitzähler
0712	CA 06		ST 06(2)	Lese-Bit selektieren
0714	C4 02		LDI #02	Lese-Strobe ausgeben
0716	07		CAS	
0717	06		CSA	EX, Bit5 = gelesenes Bit
0718	D4 20		ANI #20	
071A	01		XAE	
071B	C4 00		LDI #00	Datenport = 0
071D	CA 06		ST 06(2)	
071F	07		CAS	Lese-Strobe = 0
0720	01		XAE	A=gelesenes Bit, EX=0
0721	98 B4		JZ L06D9	Sprung wenn Bit=0 (= erfolgreich)
Bit brennen nicht erfolgreich				
0723	C1 19		LD KYLI 2(1)	A=Brennzähler
0725	D4 80		ANI #80	8 Versuche?
0727	98 C3		JZ L06ED	Nein, nochmals
Fehlermeldung ausgeben, <Adresse> EE: P3=Fehleradresse				
0729	C4 01		LDI #01	
072B	37		XPAH3	
072C	C9 12		ST DSADH(1)	
072E	C4 0F		LDI #0F	
0730	33		XPAL3	
0731	C9 13		ST DSADL(1)	
0733	C4 EE		LDI #EE	
0735	C9 14		ST DSCOD(1)	
0737	3F		XPPC3	Sprung nach 0110, EINAU
Rücksprung ins Betriebsprogramm				
0738	C4 00		LDI #00	Sprung nach 0000, Reset
073A	37		XPAH3	

Address	Code	Label	Mnemonic	Comment
073B	C4 00		LDI #00	
073D	33		XPAL3	
073E	3F		XPPC3	
073F	00		HALT	Unbenutzter Bereich
0740	00		HALT	
0741	00		HALT	
0742	00		HALT	
0743	00		HALT	

Taste PC: Displacement-Berechnung

0744	C4 00	L0744:	LDI #00	Anzeige Adresse = A
0746	C9 12	L0746:	ST DSADH(1)	
0748	C9 13		ST DSADL(1)	
074A	C9 14		ST DSCOD(1)	
074C	C9 10	L074C:	ST ISEDH(1)	Eingabe SED = A, Wieder-Einsprung nach Berechnung
074E	C9 11		ST ISEDL(1)	
0750	C4 01	L0750:	LDI #01	Call EINAU (0120)
0752	37		XPAH3	
0753	C4 0F		LDI #0F	
0755	33		XPAL3	
0756	3F		XPPC3	
0757	C1 10		LD ISEDH(1)	Eingabe nach Anzeige Adresse
0759	C9 12		ST DSADH(1)	
075B	C1 11		LD ISEDL(1)	
075D	C9 13		ST DSADL(1)	Keine Funktionstaste, Eingabe rotierend anzeigen und erneut lesen
075F	C1 16		LD SPKEY(1)	
0761	98 ED		JZ L0750	
0763	E4 0E		XRI #KEYIN	IN-Taste?
0765	9C 0A		JNZ L0771	Nein, andere Funktionstaste
0767	C1 12		LD DSADH(1)	IN-Taste gedrückt: Startadresse nach PC Buffer
0769	C9 00		ST BFPCH(1)	
076B	C1 13		LD DSADL(1)	
076D	C9 01		ST BFPCL(1)	
076F	90 D3		JMP L0744	
0771	C1 12	L0771:	LD DSADH(1)	Zieladresse nach P1 Buffer
0773	C9 02		ST BFP1H(1)	
0775	C1 13		LD DSADL(1)	
0777	C9 03		ST BFP1L(1)	Displacement berechnen, Resultat steht in DisplayAdresse und DisplayCode
0779	03		SCL	
077A	C1 03		LD BFP1L(1)	
077C	F9 01		CAD BFPCL(1)	
077E	C9 13		ST DSADL(1)	
0780	C9 14		ST DSCOD(1)	
0782	C9 04		ST BFP2H(1)	
0784	C1 02		LD BFP1H(1)	
0786	F9 00		CAD BFPCH(1)	
0788	C9 12		ST DSADH(1)	
078A	06		CSA	Sense A aktiv (Drucker angeschlossen)?
078B	D4 10		ANI #10	
078D	98 BD	DONE:	JZ L074C	Nein, Sprung zu neuer Berechnung

Resultat im Format [Start] -> [Ziel] : V|R|F [Displacement] ausdrucken (F=Fehler, kein Displ.)

Address	Code	Label	Mnemonic	Comment
078F	C4 07	L078F:	LDI #07	Sense A aktiv (Drucker erkannt): Call 07FE (JMP BYT2HEX, PC Buffer HOB (=Startadresse) nach ASCII hex wandeln und in Druckbuffer (05EA) schreiben
0791	37		XPAH3	
0792	C4 FD		LDI #FD	
0794	33		XPAL3	
0795	C4 05		LDI #05	
0797	36		XPAH2	
0798	C4 E0		LDI #EA	
079A	32		XPAL2	
079B	3F		XPPC3	
079C	3F		XPPC3	LOB Startadresse -> Druckbuffer
079D	C6 01		LD @01 (2)	'->' in Druckbuffer schreiben
079F	C4 2D		LDI #2D	
07A1	CE 01		ST @01 (2)	
07A3	C4 3E		LDI #3E	
07A5	CE 02		ST @02 (2)	
07A7	3F		XPPC3	HOB Zieladresse -> Druckbuffer
07A8	3F		XPPC3	LOB Zieladresse -> Druckbuffer
07A9	C6 01		LD @01 (2)	': ' -> Druckbuffer
07AB	C4 3A		LDI #3A	
07AD	CE 02		ST @02 (2)	
07AF	C1 0E		LD 0E(1)	P1=0404 (BYT2HEX), EA=0412, DSADH
07B1	E4 FF		XRI #FF	A=FF?
07B3	9C 06		JNZ L07BB	Nein, Test auf Vorwärts
07B5	C4 52		LDI #52	'R ' -> Druckbuffer (= 'Retour')
07B7	CE 02		ST @02 (2)	
07B9	90 12		JMP L07CD	Displacement -> Druckbuffer
07BB	C1 0E	L07BB:	LD 0E(1)	P1=0404 (BYT2HEX), EA=0412, DSADH
07BD	98 0A		JZ L07C9	Wenn A = 0, Vorwärts, sonst Fehler
Fehler in Eingabe, Bereichsüberlauf, Ausdruck [Start] -> [Ende] : F				
07BF	C4 46	L07BF:	LDI #46	'F' (= 'Fehler'), Bereichsüberlauf
07C1	CA 00		ST 00(2)	
07C3	C4 00		LDI #00	
07C5	C9 10		ST ISEDH(1)	
07C7	90 05		JMP PRINT	Sprung zu AUSDR
07C9	C4 56	L07C9:	LDI #56	'V ' -> Druckbuffer (= 'V'orwärts)
07CB	CE 02		ST @02 (2)	
07CD	3F	L07CD:	XPPC3	Displacement -> Druckbuffer
Zeile ausdrucken				
07CE	C4 02	PRINT:	LDI #02	Call 02B0, AUSDR, P2 bleibt unverändert
07D0	37		XPAH3	
07D1	C4 AF		LDI #AF	
07D3	33		XPAL3	
07D4	3F		XPPC3	
07D5	C4 00		LDI #00	Sprung zu neuer Berechnung
07D7	90 B4		JMP DONE	

Address	Code	Label	Mnemonic	Comment
Binär ((P1), oberes Nibble) zu ASCII-HEX Wandlung (Aufruf über XPPC3)				
07D9	C4 01	BYT2HEX:	LDI #01	F0=1, Flag für ???
07DB	07		CAS	
07DC	C1 00		LD BFPCH(1)	A=(P1), zu Beginn HOB Startadresse
07DE	1C		SR	A >>= 4
07DF	1C		SR	
07E0	1C		SR	
07E1	1C		SR	
Binär (A, unteres Nibble) zu ASCII-HEX Wandlung				
07E2	02	NI B2HEX:	CCL	
07E3	F4 F6		ADI #F6	Test ob A > 9
07E5	94 04		JP L07EB	Sprung wenn > 9, 40 addieren
07E7	F4 3A		ADI #3A	A<=9, 3A addieren
07E9	90 02		JMP L07ED	
07EB	F4 40	L07EB:	ADI #40	CL/Y = 1 (ADI #F6), 41 wird addiert
07ED	CE 01	L07ED:	ST @01(2)	Zeichen in (Druckbuffer++)
07EF				
07EF	06		CSA	F0 = 1?
07F0	D4 01		ANI #01	
07F2	98 09		JZ L07FD	Nein, Wandlung fertig, Rücksprung
07F4				
07F4	C4 00		LDI #00	F0 = 0, nach hex-Wandlung nach 07FD
07F6	07		CAS	
07F7	C5 01		LD @01(1)	Nächstes Byte aus Eingabe Buffer, P1++
07F9	D4 0F		ANI #0F	
07FB	90 E5		JMP NI B2HEX	Sprung zu binär-hex Wandlung, Nibble
07FD				
07FD	3F	L07FD:	XPPC3	Wandlung fertig, Rücksprung (RET)
07FE	90 D9	L07FE:	JMP BYT2HEX	Einsprung zu bin-hex Wandlung, Byte